Example of a BLOCK–CHARGE Time of Use Rate

Time of use rates are somewhat more complex because energy and/or demand charges vary according to the time of day and day of week. Accordingly, the BLOCK–SCH must be defined to switch from one BLOCK–CHARGE to another on an hourly basis. The following is an example of a time of use rate:

```
TIME-OF-USE = UTILITY-RATE
              RESOURCE          = ELECTRICITY
              BILLING-DAYS      = (31)
              BLOCK-CHARGES     = (WIN-PK,WIN-OFFPK,
                                =  SUM-PK,SUM-OFFPK)   ..

WIN-PK      = BLOCK-CHARGE
              BLOCK-SCH         =  SCH-BLOCK
              SCH-FLAG          =  1.2
              BLOCK1-TYPE       =  KWH/KW
                                  $SIZE COST LIMIT
              BLOCK1-DATA       =  (100, .05,  0,
                                      0, .04,  0)   ..

WIN-OFFPK   = BLOCK-CHARGE
              BLOCK-SCH         =  SCH-BLOCK
              SCH-FLAG          =  1.1
              BLOCK1-TYPE       =  KWH/KW
                                  $SIZE COST LIMIT
              BLOCK1-DATA       =  (100, .04,  0,
                                      0, .03,  0)   ..

SUM-PK      = BLOCK-CHARGE
              BLOCK-SCH         =  SCH-BLOCK
              SCH-FLAG          =  2.2
              BLOCK1-TYPE       =  KWH/KW
                                  $SIZE COST LIMIT
              BLOCK1-DATA       =  ( 100,.09,  0,
                                       0,..08,  0)   ..
```

```
SUM-OFFPK = BLOCK-CHARGE
            BLOCK-SCH       =  SCH-BLOCK
            SCH-FLAG        =  2.1
            BLOCK1-TYPE     =  KWH/KW
                                 $SIZE COST LIMIT
            BLOCK1-DATA     =  (100, .05,   0,
                                   0, .04,   0)  ..

SCH-BLOCK = SCHEDULE  THRU APR 30 (WD)    (1,6)  (1.1)
                                         (7,18)  (1.2)
                                        (19,24)  (1.1)
                                  (WEH)  (1,24)  (1.1)
                      THRU OCT 30 (WD)   (1,12)  (2.1)
                                        (12,18)  (2.2)
                                        (19,24)  (2.1)
                                  (WEH)  (1,24)  (2.1)
                      THRU DEC 31 (WD)    (1,6)  (1.1)
                                         (7,18)  (1.2)
                                        (19,24)  (1.1)
                                  (WEH)  (1,24)  (1.1)  ..
```

For a customer to utilize a time of use rate, the utility must provide a meter which is capable of recording the distribution of energy (and demand) consumption, not just the total amount used in the billing period. DOE-2 reflects this; in the above example, the costs for each block charge will be computed using only the energy consumed during the period defined by the BLOCK-SCH.

In the above example, energy costs are computed using kWh/kW blocks. The actual size of the block is therefore based on demand. By default, the demand used for each BLOCK-CHARGE is the maximum demand encountered during the block's active period, as defined by its BLOCK-SCH.

In the above time of use examples, the BILLING-DAYS and the SCH-BLOCK were defined so that the winter season changed to summer on the billing day. What happens when the season changes in the middle of the billing period? Usually, the utility will compute the charges for each peak (or off-peak) BLOCK-CHARGE using the energy consumed during the entire on-peak time of the billing period (i.e., the energy used in the computation for each on-peak BLOCK-CHARGE is the sum of the energy used in both the winter and summer on-peak blocks). The utility will then prorate the costs between the winter and summer BLOCK-CHARGEs as described previously.

For this seasonal change to computed properly, how does DOE-2 know which summer and winter blocks share the on-peak (or mid-peak, off-peak, etc.) periods? It does this through the TOU-SEASON-LINK keyword. The following example illustrates how seasonal blocks can be linked together. This is the same example as before, except that the BILLING-DAYS = (15) so that the billing day and the seasonal change no longer coincide. For clarity, the changes have been underlined:

```
TIME-OF-USE = UTILITY-RATE
              RESOURCE          =  ELECTRICITY
              BILLING-DAYS      =  (15)
              ----------------------
              BLOCK-CHARGES     =  (WIN-PK,WIN-OFFPK,
                                    SUM-PK,SUM-OFFPK)   ..

WIN-PK      = BLOCK-CHARGE
              BLOCK-SCH         =  SCH-BLOCK
              SCH-FLAG          =  1.2
              TOU-SEASON-LINKS  =  (SUM-PK)
              ----------------------------
              BLOCK1-TYPE       =  KWH/KW
                                  $SIZE  COST  LIMIT
              BLOCK1-DATA       =  ( 100,  .05,   0,
                                       0,  .04,   0)   ..

WIN-OFFPK   = BLOCK-CHARGE
              BLOCK-SCH         =  SCH-BLOCK
              SCH-FLAG          =  1.1
              TOU-SEASON-LINKS  =  (SUM-OFFPK)
              ----------------------------
              BLOCK1-TYPE       =  KWH/KW
                                   $SIZE  COST  LIMIT
              BLOCK1-DATA       = ( 100,  .04,   0,
                                       0,  .03,   0)   ..

SUM-PK      = BLOCK-CHARGE
              BLOCK-SCH         =  SCH-BLOCK
              SCH-FLAG          =  2.2
              TOU-SEASON-LINKS  =  (WIN-PK)
              ----------------------------
              BLOCK1-TYPE       =  KWH/KW
                                  $ SIZE  COST  LIMIT
              BLOCK1-DATA       =  ( 100,  .09,   0,
                                       0,  .08,   0)   ..
```

```
SUM-OFFPK    = BLOCK-CHARGE
               BLOCK-SCH          =  SCH-BLOCK
               SCH-FLAG           =  2.1
               TOU-SEASON-LINKS   =  (WIN-OFFPK)
               ----------------------------------
               BLOCK1-TYPE        =  KWH/KW
                                     $ SIZE   COST   LIMIT
               BLOCK1-DATA        =  ( 100,   .05,    0,
                                       0,    .04,    0)   ..

SCH-BLOCK = SCHEDULE  THRU APR 30 (WD)    (1,6)  (1.1)
                                         (7,18)  (1.2)
                                        (19,24)  (1.1)
                                  (WEH)  (1,24)  (1.1)
                      THRU OCT 30 (WD)   (1,12)  (2.1)
                                        (12,18)  (2.2)
                                        (19,24)  (2.1)
                                  (WEH)  (1,24)  (2.1)
                      THRU DEC 31 (WD)    (1,6)  (1.1)
                                         (7,18)  (1.2)
                                        (19,24)  (1.1)
                                  (WEH)  (1,24)  (1.1)   ..
```

To summarize, the TOU-SEASON-LINKS keyword is required only when a time of use rate is being simulated and the BILLING-DAYS does not coincide with the change in season. In this case, report ES-F will report the "metered energy" for each BLOCK-CHARGE as the energy metered during the period defined by the BLOCK-SCH, the "billing energy" as the sum of the energy metered for this block and its linked block, and the "prorate factor" as the number of hours that this block was active relative to its linked block (i.e., prorate factor = Hours1/(Hours1+Hours2) where Hours1 is the number of active hours of this block, and Hours2 is the number of active hours of the linked block). As for a non time of use seasonal change, the prorate factors of two linked blocks should always add up to 1.0

How does DOE-2 know whether a BLOCK-CHARGE is being used in a yearly, seasonal, or time of use format? It does this by looking at the number of times the BLOCK-SCH changes during the course of the year. If the schedule never changes, the block must be yearly. If it changes no more than once in each billing period, it is seasonal. If it changes more than once in any billing period, it is considered to be a time of use block. Report ES-F indicates whether each BLOCK-CHARGE is yearly, seasonal, or time of use. You should always review this report to confirm that the program is modeling your BLOCK-CHARGEs as intended.

SCHEDULE

This command, along with DAY-SCHEDULE and WEEK-SCHEDULE, is used to coordinate the operation of the UTILITY-RATEs, BLOCK-CHARGEs and RATCHETs. Because a utility's rate structure may be complex, a SCHEDULE may need to coordinate a large number of different items. For this purpose, SCHEDULEs may be provided with user-defined flag values which are used to activate different rates, blocks, or ratchets at different times of the day or season.

You are already familiar with schedule flags. For example, the FAN-SCHEDULE in SYSTEMS uses the flag values of 0 and 1; 0 means the fan is off, and 1 means the fan is on. In ECONOMICS, 1 may represent the winter season, and 2 the summer. Similarly, for time of use demand pricing, 1.1, 1.2 and 1.3 may represent the peak, shoulder, and off-peak demand periods in winter, while 2.1, 2.2, and 2.3 are for the summer. Using flags, a time of use demand schedule can be defined as follows.

```
TOU-SCHEDULE = SCHEDULE   THRU MAR 31   (WD)     (1,6)  (1.3)
                                                 (7,12)  (1.2)
                                                (13,18)  (1.1)
                                                (19,24)  (1.3)
                                        (WEH)    (1,24)  (1.3)

                          THRU OCT 15   (WD)     (1,6)  (2.3)
                                                 (7,12)  (2.2)
                                                (13,18)  (2.1)
                                                (19,24)  (2.3)
                                        (WEH)    (1,24)  (2.3)

                          THRU DEC 31   (WD)     (1,6)  (1.3)
                                                 (7,12)  (1.2)
                                                (13,18)  (1.1)
                                                (19,24)  (1.3)
                                        (WEH)    (1,24)  (1.3)  ..
```

Each BLOCK-CHARGE associated with one of these periods references the schedule, and also references a specific flag. For example, the BLOCK-CHARGE defined with the peak summer period would reference the flag value 2.1. Note that you define the flag values; any numbers acceptable by the SCHEDULE commands are acceptable.

Examples of Electricity Tariffs
To illustrate the use of the ECONOMICS commands and keywords, a series of examples of electricity tariffs is presented. These examples can be extended to other fuels and utilities.

Example 1: Basic Tariff
The most basic tariff is a uniform charge levied on all units consumed in a month. For this example, all kilowatt-hours cost $0.05 and there is a monthly customer charge of $15.00. The minimum bill is $17.00 and there are no demand charges.

```
ELEC-TARIFF = UTILITY-RATE
                RESOURCE       = ELECTRICITY   $required
                MONTH-CHGS     = (15.)
                ENERGY-COST    = .05
                MIN-MON-CHGS   = (17.)   ..
```

MONTH-CHGS and MIN-MON-CHGS take lists specifying the charges for 12 months. Since only a single value was entered, this value will be used for all 12 months.

Example 2: Simple Block Tariff
Although block rates have been used for years, many of them now incorporate marginal-cost and equity-related concerns. A recent example of the latter, currently in wide usage among residential customers, are inverted block rates. The basic idea is that increased consumption is discouraged by increased per unit costs. A simple inverted block has three tiers. In this example, the first 500 kWh of consumption (sometimes referred to a "baseline" or "life line" quantity) are charged at $.0535 per kWh. All kWh consumed in excess of 500 kWh, but less than 900 kWh, are charged at $.0725 per kWh. The third tier covers all consumption in excess of 900 kWh at a charge of $.1245 per kWh. There is no seasonal variation in this rate and we will ignore minimum and fixed monthly charges in this example.

```
ELEC-TARIFF = UTILITY-RATE   RESOURCE       = ELECTRICITY $required
                             BLOCK-CHARGES  = (INVBLK)   ..

INVBLK      = BLOCK-CHARGE   BLOCK1-TYPE    = ENERGY
                                 $SIZE   COST
                             BLOCK1-DATA    = (500,  .0535,
                                               400,  .0725,
                                                 1,  .1245 )   ..
```

Note that the size of the last block can be any number. Since BLOCK1-DATA is not followed by BLOCK2-DATA, all remaining energy will be assessed at the rate in the last set.

Example 3: Seasonal Block Tariff

Most utilities are faced with demands for electricity that are not evenly distributed throughout the year. They reflect the fact that changing levels of demand result in differing costs of service by introducing seasonal variations in the rates for electricity. These variations may have different size blocks associated with them, as well. In this next example, there is a winter season that lasts from October to May and a summer season that lasts from June to September. This utility is winter-peaking, but recognizes the need for increased life line allowances at this time of year.

```
ELEC-TARIFF = UTILITY-RATE
              RESOURCE       =  ELECTRICITY   $REQUIRED
              BLOCK-CHARGES  =  (WINTER-BLK,  SUMMER-BLK)   ..


WINTER-BLK = BLOCK-CHARGE
             BLOCK-SCH      =  SEASONS-SCH
             SCH-FLAG       =  1
             BLOCK1-TYPE    =  ENERGY   $REQUIRED
                            $ SIZE   COST
             BLOCK1-DATA    =  ( 1000,.07,
                                    1,  .10 )   ..


SUMMER-BLK = BLOCK-CHARGE
             BLOCK-SCH   =  SEASONS-SCH
             SCH-FLAG    =  2
             BLOCK1-TYPE =  ENERGY
                         $ SIZE COST
             BLOCK1-DATA =  (   500,.06,
                                  1,.09)   ..


SEASONS-SCH = SCHEDULE   THRU MAY 31   (ALL)   (1,24)  (1)
                         THRU SEP 30   (ALL)   (1,24)  (2)
                         THRU DEC 31   (ALL)   (1,24)  (1)   ..
```

Note how the use of the SCH-FLAG keyword allowed both BLOCK-CHARGEs to reference the same schedule.

Example 4: Demand Charges

The most significant difference between residential and commercial electricity tariffs is the inclusion of demand charges. These tariffs can also include rate limitation features to ensure that when the charges are all totaled, the effective rate per kWh is less than or equal to a specified amount. In this example the demand charge is \$12/kW. There is a flat charge on energy of \$0.05/kWh, but in no circumstance can the effective rate (i.e., including the demand charges) exceed \$.07/kWh.

```
ELEC-TARIFF  =  UTILITY-RATE
                RESOURCE          = ELECTRICITY
                ENERGY-CHGS       = (0.05)
                DEMAND-CHGS       = (12.00)
                RATE-LIMITATION = 0.07  ..
```

Example 5: Time of Use and Seasonal Demand Charges

The most recent innovation in rate design has been the introduction of time of use rates wherein the time of day, week, and year that energy is consumed get broken into different costing periods and have different charges assigned to them. The charges, moreover, can be for demand and energy, and for each of these the definition of the periods can change. In this example, there is a winter and summer season. Energy charges vary by season and also by on-peak and off-peak. Demand charges vary by season only, and are charged at \$5.00/kW during the winter (Oct-Mar) and \$8.00/kW during the summer.

```
TIME-OF-USE  =  UTILITY-RATE
                RESOURCE          = ELECTRICITY
                ENERGY-CHG-SCH    = ENERGY-SCH
                DEMAND-CHGS       = (5,5,5,8,8,8,8,8,8,
                                     5,5,5)  ..

ENERGY-SCH  = SCHEDULE     THRU MAR 31 (WD)     (1,8)  (.04)
                                               (9,22)  (.06)
                                              (23,24)  (.04)
                                    (WEH)     (1,24)  (.04)
                           THRU SEP 31 (WD)     (1,8)  (.05)
                                               (9,20)  (.07)
                                              (21,24)  (.05)
                                    (WEH)     (1,24)  (.05)
                           THRU DEC 31 (WD)     (1,8)  (.04)
                                               (9,22)  (.06)
                                              (23,24)  (.04)
                                    (WEH)     (1,24)  (.04)  ..
```

Shoulder periods are those times during the day when the utility experiences moderate use; they can be easily incorporated by including additional times in the ENERGY-SCH.

ECONOMICS–REPORT

This instruction defines which ECONOMICS reports will be output. Users can select from *verification* reports and *summary* reports. Verification reports echo your input; summary reports show calculation results, usually monthly and annually.

Format:

```
ECONOMICS–REPORT    VERIFICATION = (code-word list)
                    SUMMARY = (code-word list)  ..
```

Example:

```
ECONOMICS–REPORT    VERIFICATION = (none required)
                    SUMMARY = (ES–D, ES–E)  ..
```

will print summary reports ES–D, "Summary of Fuel and Utility Use and Costs", and ES–E, "Summary of Electricity Charges". A definition of all reports, with corresponding code-words, is given in Appendix D.